

# Les sous-programmes (Algo)

## Corrigé

### Résumé

Ce document décrit les sous-programmes, outil essentiel pour structurer un programme car il permet au programmeur de définir ses propres instructions et opérateurs.

## Table des matières

|          |  |           |
|----------|--|-----------|
| <b>1</b> | <b>Introduction</b>                                      | <b>4</b>  |
| <b>2</b> | <b>Exemple introductif</b>                               | <b>5</b>  |
| 2.1      | Définition de nos premiers sous-programmes . . . . .     | 5         |
| 2.2      | Écriture de l’algorithme . . . . .                       | 6         |
| 2.3      | Déplacer un robot avec des sous-programmes . . . . .     | 7         |
| 2.4      | Exécution d’un sous-programme . . . . .                  | 8         |
| 2.5      | Procédures utilisant d’autres procédures . . . . .       | 9         |
| 2.6      | Exemples de fonctions . . . . .                          | 9         |
| 2.7      | Nombre de paramètres formels . . . . .                   | 10        |
| <b>3</b> | <b>Intérêt des sous-programmes</b>                       | <b>11</b> |
| <b>4</b> | <b>Portée des variables</b>                              | <b>13</b> |
| 4.1      | Portée des variables . . . . .                           | 13        |
| 4.2      | Variable globale . . . . .                               | 13        |
| 4.3      | Variable locale . . . . .                                | 13        |
| 4.4      | Masquage . . . . .                                       | 14        |
| 4.5      | Durée de vie d’une variable . . . . .                    | 15        |
| 4.6      | Illustration . . . . .                                   | 15        |
| <b>5</b> | <b>Structure d’un algorithme</b>                         | <b>17</b> |
| <b>6</b> | <b>Communication entre programmes et sous-programmes</b> | <b>18</b> |
| 6.1      | Les paramètres . . . . .                                 | 18        |
| 6.1.1    | Pourquoi des modes de passage ? . . . . .                | 19        |
| 6.1.2    | Quels modes de passage des paramètres ? . . . . .        | 19        |
| 6.1.3    | Paramètres en entrée . . . . .                           | 20        |
| 6.1.4    | Paramètres en sortie . . . . .                           | 21        |

|           |  |           |
|-----------|--|-----------|
| 6.1.5     | Paramètres en entrée/sortie . . . . .                      | 22        |
| 6.2       | Les variables globales . . . . .                           | 23        |
| <b>7</b>  | <b>Exécution d'un appel à un sous-programme</b>            | <b>25</b> |
| 7.1       | Vérification statique . . . . .                            | 25        |
| 7.2       | Exécution d'un appel . . . . .                             | 25        |
| 7.3       | Illustration . . . . .                                     | 26        |
| <b>8</b>  | <b>Procédures</b>  | <b>27</b> |
| 8.1       | Interface d'une procédure . . . . .                        | 27        |
| 8.1.1     | Signature . . . . .  | 27        |
| 8.1.2     | Sémantique . . . . .                                       | 28        |
| 8.2       | Implantation d'une procédure . . . . .                     | 28        |
| 8.3       | Comment définir une procédure ? . . . . .                  | 28        |
| <b>9</b>  | <b>Fonctions</b>   | <b>30</b> |
| 9.1       | Interface d'une fonction . . . . .                         | 30        |
| 9.1.1     | Signature . . . . .  | 30        |
| 9.1.2     | Sémantique . . . . .                                       | 30        |
| 9.2       | Implantation d'une fonction . . . . .                      | 30        |
| <b>10</b> | <b>Procédure ou fonction ?</b>                             | <b>31</b> |
| 10.1      | La fonction comme procédure . . . . .                      | 31        |
| 10.2      | Intérêt des fonctions par rapport aux procédures . . . . . | 31        |
| 10.3      | Choisir entre fonction et procédure . . . . .              | 32        |
| <b>11</b> | <b>Liens entre raffinages et sous-programmes</b>           | <b>33</b> |
| <b>12</b> | <b>Imbrication de sous-programmes</b>                      | <b>34</b> |
| 12.1      | Tolérance sur les sous-programmes . . . . .                | 34        |
| 12.2      | Sous-programmes d'intérêt général . . . . .                | 34        |
| 12.2.1    | Sous-programmes et langages de programmation . . . . .     | 36        |

## Liste des exercices

Exercice 1 : Sous-programmes déjà rencontrés ..... 4

# 1 Introduction

**Motivation :** Un sous-programme permet au programmeur d'étendre le langage de programmation en définissant ses propres instructions et ses propres opérateurs qu'il pourra ensuite réutiliser. Il s'agit de regrouper plusieurs instructions sous un nom commun.

**Définition :** Un sous-programme est un programme indépendant de son contexte (autonome), réutilisable dans plusieurs contextes.

**Terminologie :** Un *sous-programme* est un regroupement d'instructions qui peuvent ensuite être exécutées de manière atomique. Un sous programme permet ainsi de définir des instructions ou des expressions utilisateurs. Dans le premier cas on parle de procédure et dans le second cas de fonction.

Le terme « sous-programme » est un terme générique qui englobe à la fois les **procédures** et les **fonctions**. Les procédures et les fonctions peuvent être respectivement considérées comme des instructions et des opérateurs (ou expressions) définis par le programmeur qu'il pourra ensuite utiliser comme si elles faisaient partie du langage, donc comme des instructions élémentaires.

Les notions de procédure et de fonction sont très proches. La principale différence est qu'une procédure réalise un traitement et est assimilable à une instruction alors qu'une fonction (calcule et) renvoie une information et correspond donc à une expression. Un appel à une fonction peut apparaître là où une expression peut être utilisée (à droite de l'affectation, dans une expression, dans une condition, etc.)

Dans la section 2, nous introduisons les notions de procédure et de fonction en nous appuyant sur des exemples concrets. Dans les sections suivantes, nous définissons plus précisément ce que sont une procédure et une fonction.

## Exercice 1 : Sous-programmes déjà rencontrés

Donner des exemples de sous-programmes que vous avez déjà rencontré. Préciser si ce sont des procédures ou des fonctions.

**Solution :** Lire et Écrire sont des exemples de procédure.  
abs est un exemple de fonction

**Conclusion :** On a déjà rencontré et utilisé des sous-programmes.

## 2 Exemple introductif

### 2.1 Définition de nos premiers sous-programmes

Reprenons l'exemple du robot. Dans les niveaux de raffinement intermédiaires nous avons identifié des étapes telles que « tourner à gauche », « tourner à droite », « progresser de  $n$  cases »,  $n$  étant un entier positif.

Puisque ces étapes ne sont pas comprises par le robot, il était nécessaire dans l'algorithme final de remplacer chacune de leurs occurrences par les instructions élémentaires correspondantes. Ceci est fastidieux (on répète les mêmes instructions), conduit à un algorithme long et difficile à comprendre (il faut absolument mettre les commentaires correspondant aux traces du raffinement).

Les procédures (et fonctions) proposent une solution élégante et économique. Nous allons définir ces trois étapes comme étant des procédures (des instructions programmeur) en expliquant une fois pour toutes au processeur comment les comprendre (les interpréter, les exécuter). Elles constituent donc de nouvelles instructions que nous pourrions ensuite utiliser pour guider le robot.

Comment peut-on définir de telles opérations ?

1. On commence par préciser ce que doit faire le sous-programme en définissant son interface qui comprend, entre autre, le nom du sous-programme, son type (ici procédure) et un commentaire qui décrit son objectif informellement.
2. On décrit comment réaliser le traitement décrit dans l'interface. Ceci correspond à un algorithme obtenu de la même manière que pour un programme dont le R0 serait l'interface du sous-programme.

```

1 Procédure tourner_droite Est
2   -- faire tourner le robot à droite
3   Début
4     PIVOTER
5   Fin
6
7 Procédure tourner_gauche Est
8   -- faire tourner le robot à gauche
9   Début
10    PIVOTER
11    PIVOTER
12    PIVOTER
13   Fin

```

**Remarque :** Si on suppose qu'on a la notion de compteur (entier, incrémentation et test) alors on peut écrire la procédure de cette manière.

```

1 Procédure tourner_gauche Est
2   -- faire tourner le robot à gauche
3   Variable
4     i: Entier      -- compter le nombre de fois où l'on tourne
5   Début
6     Pour i <- 1 Jusqu'à i = 3 Faire
7       PIVOTER
8   FinPour

```

9 **Fin**

Dans ce cas, pour réaliser la procédure, on utilise une variable *i*. Cette variable est dite **variable locale** à la procédure car elle est seulement utilisée dans le traitement réalisé par la procédure. Le programme appelant cette procédure ne connaît pas (et n'a pas à connaître) l'existence de cette variable. Ainsi :

- sa **durée de vie** est celle du sous-programme (elle est créée quand le sous programme est appelé et est détruite lorsqu'il se termine : c'est l'**allocation statique** de mémoire réalisée automatiquement par le compilateur).
- sa **portée** est limitée aux instructions du sous-programme. La variable n'est visible, et donc ne peut être utilisée, que dans le sous-programme.

De la même manière on peut définir la procédure progresser.

```

1 Procédure progresser(n : in Entier)
2     -- Faire avancer le robot de n cases dans la direction courante.
3     --
4     -- Nécessite :
5     --     n > 0     -- n doit être un entier strictement positif
6     Variable
7     i : Entier
8     Début
9         Pour i <- 1 JusquÀ i = n Faire
10            AVANCER
11        FinPour
12    Fin

```

La procédure progresser ainsi définie possède un paramètre formel nommé *n*. Chaque paramètre formel correspond à une information échangée entre le programme appelant et le sous-programme. Il est caractérisé par un nom, un type et un mode de passage qui décrit dans quel sens la communication a lieu (section 6.1). Ici le mode de passage est **in** (en entrée), ce qui signifie que c'est au programme appelant de fournir la valeur de *n* et que le sous-programme n'a pas le droit d'en changer la valeur. La valeur fournie par le programme appelant est appelée le *paramètre effectif*.

Ainsi, lorsque l'on écrit un sous-programme, on raisonne sur la valeur des paramètres formels, sachant que les valeurs effectives (réelles) ne sont pas encore connues. Dans le cas, présent on pourrait formuler le sous-programme de la manière suivante : « étant donné un entier *n*, faire avancer le robot de *n* cases dans la direction courante ».

La clause Nécessite qui fait partie du commentaire indique les conditions que doivent vérifier les paramètres effectifs fournis par le programme appelant. Ici, il faut nécessairement donner une valeur strictement positive pour le paramètre formel *n*. Si ceci n'est pas respecté, c'est le programmeur du programme appelant qui est en faute et non l'auteur du sous-programme progresser.

## 2.2 Écriture de l'algorithme

Maintenant que les sous-programmes précédents sont écrits, l'algorithme qui consiste à déplacer le robot de la salle de cours vers le secrétariat s'écrit ainsi.

```

1  Algorithme Guider_robot
2
3      -- Guider le robot de la salle de cours vers le secrétariat
4
5      -- la définition des sous-programmes est omise.
6
7  Début
8      -- Sortir de la salle de cours
9      progresser(2)
10     tourner_gauche
11     progresser(3)
12
13     -- Longer le couloir (jusqu'à la porte du vestibule)
14     tourner_droite
15     progresser(9)
16
17     -- Traverser le vestibule (et entrer dans le secrétariat)
18     tourner_droite
19     progresser(3)
20     tourner_droite
21     progresser(1)
22     tourner_gauche
23     progresser(1)
24 Fin.

```

## 2.3 Déplacer un robot avec des sous-programmes

Voici l'algorithme complet du déplacement du robot faisant apparaître les sous-programmes définis ci-dessus et le programme principal.

```

1  Algorithme Guider_robot
2
3      -- Guider le robot de la salle de cours vers le secrétariat
4      -- Solution algorithmique utilisant les sous-programmes.
5
6  Procédure tourner_droite Est
7      -- Faire tourner le robot à droite
8      Début
9          PIVOTER
10     Fin
11
12 Procédure tourner_gauche Est
13     -- Faire tourner le robot à gauche
14     Début
15         PIVOTER
16         PIVOTER
17         PIVOTER
18     Fin
19
20 Procédure progresser(n: in Entier) Est
21     -- Faire avancer le robot de n cases dans la direction courante.

```

```

22      --
23      -- Nécessite :
24      --      n > 0    -- n doit être un entier strictement positif
25  Variables
26      i: Entier      -- pour compter le nombre de fois où le robot avance
27  Début
28      Pour i <- 1 Jusqu'À i = N Faire
29          AVANCER
30      FinPour
31  Fin
32
33  ----- L'algorithme principal -----
34
35  Début
36      -- Sortir de la salle de cours
37      progresser(2)
38      tourner_gauche
39      progresser(3)
40
41      -- Longer le couloir (jusqu'à la porte du vestibule)
42      tourner_droite
43      progresser(9)
44
45      -- Traverser le vestibule (et entrer dans le secrétariat)
46      tourner_droite
47      progresser(3)
48      tourner_droite
49      progresser(1)
50      tourner_gauche
51      progresser(1)
52  Fin.
53
54  -- Remarque : on aurait pu remplacer progresser(1) par AVANCER. Ceci améliore
55  -- l'efficacité (on évite un appel à un sous-programme) mais nuit peut-être à
56  -- l'homogénéité de l'algorithme et donc à sa clarté. Un autre avantage à utiliser
57  -- progresser(1) est que l'algorithme dépend alors seulement des procédures définies e
58  -- pas directement des actions élémentaires. Il est donc plus stable à un changement
59  -- les instructions élémentaires du robot.

```

## 2.4 Exécution d'un sous-programme

Cet aspect est abordé en détail à la section 7. Nous nous contentons ici d'en esquisser le principe.

L'appel d'un sous-programme correspond à un déroutement de l'exécution du programme. L'instruction suivante à exécuter est alors la première instruction du sous-programme. Il s'exécute comme un sous-programme mais ne peut manipuler que les données qui correspondent à ses paramètres formels (initialisés à partir des paramètres effectifs) et ses variables locales. Lors de la fin de l'exécution du sous-programme, l'exécution du programme appelant reprend à l'instruction qui suit l'appel au sous-programme. Il est donc nécessaire de savoir qu'elle est l'instruction



du programme appelant qui a provoqué l'appel. Cette information doit être conservé lors de l'appel du sous-programme. Elle est stockée dans la pile d'exécution avec les paramètres et variables locales du sous-programme, l'ensemble constituant le bloc d'activation.

## 2.5 Procédures utilisant d'autres procédures

Le corps d'une procédure est composé d'instructions qui peuvent être élémentaires ou définies par le programmeur (sous-programmes).

Ainsi, on peut définir une procédure reculer de la manière suivante :

```

1  Procédure reculer(n: in Entier) Est      -- n : paramètre formel de reculer
2      -- faire avancer le robot de n cases vers l'arrière, n étant
3      -- strictement positif. La direction du robot n'est pas modifiée.
4      --
5      -- Principe : avancer dans la direction opposée.
6      Début
7          faire_demi_tour
8          progresser(n)          -- n est le paramètre effectif de progresser
9          faire_demi_tour
10     Fin

```

Il faut bien évidemment avoir défini **au préalable** la procédure faire\_demi\_tour. Une définition possible est donnée ci-dessous.

```

1  Procédure faire_demi_tour Est
2      -- faire faire demi tour au robot
3      Début
4          PIVOTER
5          PIVOTER
6      Fin

```

En effet, on ne doit pas avoir de référence en avant car elles nuisent à la bonne compréhension de l'algorithme : il faut toujours avoir défini les entités utilisées. Une deuxième raison (certainement la plus importante) est que ceci facilite l'écriture des compilateurs.

## 2.6 Exemples de fonctions

Une fonction est une expression définie par le programmeur. Elle renvoie donc une valeur. Aussi, son identifiant ne sera pas un verbe mais un nom commun ou un adjectif si la valeur est booléenne (somme, max, bissextile, égales, est\_opération, etc.) qui caractérise l'information renvoyée. Par exemple, nous pouvons appeler carré la fonction qui associe à un réel son carré.

```

1  Fonction carré(x : in Réel) : Réel Est
2      -- le carré d'un nombre réel
3      Début
4          Résultat <- x * x
5      Fin

```

Comme une procédure, une fonction est définie par un identifiant (son nom), un commentaire expliquant ce qu'elle renvoie, des paramètres formels et un corps qui calcule le résultat à renvoyer. Lorsqu'on définit une fonction il faut en plus préciser le type de l'information renvoyée. C'est le **type de retour** de la fonction.

En algorithmique, nous considérerons que pour chaque fonction, une variable de même type que le type de retour de la fonction est définie. Le nom de cette variable est **Résultat**. Elle peut être utilisée pour réaliser les calculs. La valeur renvoyée à la fin de l'exécution de la fonction est la valeur de la variable **Résultat**.

**Remarque :** Une fonction n'est pas forcément une fonction mathématique. Exemple : la fonction qui indique si un caractère correspond à une opération (+ - \* /), la fonction qui indique si l'utilisateur répond oui ou non...

## 2.7 Nombre de paramètres formels

Un sous-programme peut ne pas avoir de paramètres (tourner\_droite et tourner\_gauche), avoir un paramètre (progresser prend en paramètre un nombre de cases), avoir plusieurs paramètres (max a deux paramètres correspondant à ses deux opérandes). Les paramètres formels sont séparés par des points virgules.

```
1  Fonction max(a, b : in Entier) : Entier Est
2      -- le plus grand de deux nombres entiers
3      --
4      -- Assure
5      --      (Résultat >= a) Et (Résultat >= b) -- plus grand que a et b
6      --      (Résultat = a) Ou (Résultat = b)   -- l'un des deux
7  Début
8      Si a > b Alors
9          Résultat <- a
10     Sinon
11         Résultat <- b
12     FinSi
13 Fin
```

Il faut éviter d'avoir trop de paramètres car il devient difficile de savoir à quoi ils correspondent. D'ailleurs certains langages limitent le nombre de paramètres.

**Remarque :** Pour diminuer le nombre de paramètres, on peut utiliser un type enregistrement. Ainsi la comparaison de deux dates (l'égalité par exemple) ne prend pas 6 paramètres (j1, m1, a1, j2, m2, a2) mais deux (d1 et d2) qui ont chacun trois champs correspondant au jour, au mois et à l'année.

### 3 Intérêt des sous-programmes

L'utilisation des sous-programmes (procédures et fonctions) est intéressante à plus d'un titre. Nous en précisons quelques uns ci-dessous.

**Structuration de l'algorithme** En l'absence de procédure et de fonction, l'algorithme est monolithique : les instructions sont toutes écrites les unes à la suite des autres et ce ne sont que les commentaires, traces du raffinages, qui le structurent et permettent de le comprendre. Chaque procédure ou fonction correspond à une sous-partie (un sous-programme) de l'algorithme initial. Il permet donc de le découper en « morceaux » et donc de le structurer en entités relativement indépendantes. Pour s'en convaincre, il suffit de regarder l'algorithme 2.3.

**Compréhensibilité de l'algorithme** La structuration permet d'améliorer la compréhension de l'algorithme, car il est découpé en procédures et fonctions de taille réduite qui sont donc plus facile à comprendre. Les algorithmes ne font donc plus nécessairement appel aux instructions élémentaires mais à des instructions de plus haut niveau les procédures et les fonctions qui, de part leur nom significatif, en améliore la compréhension.

**Factorisation du code** Les sous-programmes permettent de factoriser des parties de l'algorithme. Par exemple, tourner\_gauche remplace les trois occurrences successives de PIVOTER dont l'objectif était de faire tourner à gauche le robot. La manière de faire tourner à gauche le robot n'est donnée qu'à un seul endroit.

**Mise au point** Dès qu'un sous-programme est écrit, il peut (et doit) être testé. Ainsi, le programme est testé petits bouts par petits bouts (sous-programme après sous-programme). Les erreurs et surtout leur origine sont alors beaucoup plus facilement identifiées que si l'ensemble du l'algorithme était testé d'un seul coup.

Ceci est également vrai lors de la modification de sous-programmes (maintenance corrective ou évolutive).

**Amélioration de la maintenance** Comme la compréhension du programme, la maintenance est automatiquement améliorée, car il sera plus facile d'identifier les parties de l'algorithme à modifier et d'en évaluer l'impact. L'idéal est bien entendu que la modification puisse être limitée à un petit nombre de sous-programmes.

La factorisation améliore également notablement la maintenance. En effet, si une modification est faite dans un sous-programme, elle sera automatiquement prise en compte dans toutes les autres parties qui utilisent ce sous-programme. Si le code avait été dupliqué, il aurait fallu le changer à tous les endroits.

Dans le même ordre d'idée, les sous-programmes permettent au programmeur de définir ses propres instructions et expressions. Ainsi, s'appuyant sur ses instructions, il peut écrire des algorithmes indépendants des instructions élémentaires du processeur.

Prenons un exemple concret, les déplacements d'un robot. Tout déplacement du robot peut être fait en utilisant tourner à droite, tourner à gauche et progresser de  $n$  cases. Si les actions élémentaires du robot changent (par exemple PIVOTER ne le fait plus tourner d' $1/4$  de tour dans le sens des aiguille d'une montre mais d' $1/8$  de tour dans le sens inverse des aiguilles d'une montre), il suffit de modifier le corps (l'implantation) de ces trois procédures sans modifier le reste de l'algorithme (le déplacement de la salle de cours vers le secrétariat reste inchangé).

**Réutilisation** Un sous-programme décrit un traitement ou le calcul d'une information qui peut ensuite être utilisé autant de fois que nécessaire dans l'algorithme en cours de construction mais également dans les algorithmes futurs. Par exemple, à chaque fois qu'on doit faire tourner le robot à gauche, il suffit d'appeler la procédure `tourner_gauche`.

## 4 Portée des variables

### 4.1 Portée des variables

On appelle *portée d'une variable* la partie du programme où on peut y faire référence.

La portée d'une variable (la zone du programme dans laquelle elle peut être utilisée, on dit aussi *visible*) commence avec sa déclaration et se termine avec l'entité (programme, module, procédure, etc.) dans laquelle elle a été déclarée.

Par exemple, une variable déclarée dans un sous-programme comme le `i` de progresser ne peut être utilisée que dans le sous-programme progresser.

### 4.2 Variable globale

On appelle variable globale une variable qui est visible (et donc utilisable et modifiable) par plusieurs sous-programmes (le programme principal étant considéré comme un sous-programme).

Les variables globales sont à éviter car elles correspondent à des entrées et/ou des sorties des sous-programmes qui ne sont pas identifiées dans leurs interfaces. Elles rendent alors les programmes plus difficiles à comprendre et à maintenir (effets de bord).

### 4.3 Variable locale

Une variable locale est une variable déclarée à l'intérieur d'un sous-programme. Sa portée est donc ce sous-programme.

```

1  Algorithme ExemplePortée
2
3  Variable
4      x: Entier      -- x est une variable globale de type Entier
5
6  Procédure proc(n: in Réel) Est
7      Variable
8          i: Réel  -- la variable locale masque la variable globale i
9      Début
10         i <- n          -- OK
11         x <- 3          -- OK
12         l <- '0'        -- ERREUR : l n'est pas accessible
13     Fin
14
15 Procédure proc2(n: in Réel) Est
16     Début
17         x <- n          -- OK
18     Fin
19
20 Variable
21     l: Caractère      -- variable « locale » au programme principal
22
23 Début
24     n <- 3             -- ERREUR : n n'est pas accessible

```

```

25     i <- 1.2    -- ERREUR : i n'est pas accessible
26     x <- 1      -- OK   (x est la variable globale)
27     l <- 'A'    -- OK
28 Fin

```

Dans l'exemple ci-dessus, la portée de la variable *x* englobe la procédure *proc* et le programme principal. Il s'agit donc d'une variable globale.

En revanche la portée de *l* n'englobe que le programme principal, il s'agit donc pas d'une variable globale mais d'une variable « locale » au programme principal. Comme elle est déclarée après la procédure *proc*, elle n'est donc pas visible dans ce sous-programme ce qui explique l'erreur lors de l'affectation *l* <- '0'.

Le paramètre formel *n* et la variable locale *n* ont pour portée la procédure *proc* dans laquelle ils sont déclarés. En conséquence, ils ne peuvent pas être utilisés dans le programme principal, d'où les deux erreurs.

#### 4.4 Masquage

Pour nommer une variable locale, on peut utiliser un nom déjà utilisé pour une variable globale. La conséquence est que la nouvelle déclaration masque la précédente. Ainsi, la variable globale ne peut pas être utilisée. Son nom désignera la variable locale.

```

1 Algorithme ExempleMasquage
2
3 Variable
4   i: Réel      -- i est une variable globale de type Réel
5   n: Caractère -- n est une variable globale de type Caractère
6   x: Entier   -- x est une variable globale de type Entier
7
8 Procédure proc(n: in Réel) Est
9   -- le paramètre formel n masque la variable globale n
10  Variable
11   i: Réel    -- la variable locale masque la variable globale i
12   j: Réel    -- variable locale de type Réel
13  Début
14   i <- 5.0      -- OK (i est la variable locale)
15   j <- 5.0      -- OK (j est la variable locale)
16   n <- 1        -- OK (n est le paramètre formel)
17   x <- 3.14     -- OK (x est la variable globale)
18  Fin
19
20 Début
21   n <- 'V'      -- OK   (n est la variable globale)
22   n <- 1.5      -- ERREUR : n est de type Caractère (var. globale)
23   j <- 1.2      -- ERREUR : j n'est pas visible
24   x <- 1.3      -- OK   (x est la variable globale)
25 Fin

```

## 4.5 Durée de vie d'une variable

La durée de vie est le temps pendant lequel la variable existe, c'est-à-dire pendant lequel elle peut être utilisée. Elle commence par la réservation de sa place en mémoire et s'arrête lorsque cette place est libérée.

Une variable a la même durée de vie que l'entité dans laquelle elle a été déclarée. Ainsi, une variable du programme principal a la même durée que le programme principal : elle est créée quand le programme est lancé et est « détruite » quand le programme s'arrête. Une variable locale à un sous-programme est créée quand le sous-programme est appelé. Quand le sous-programme est terminé, elle est détruite et la place qu'elle occupait est récupérée.

## 4.6 Illustration

Illustrons ceci par un exemple. Cet exemple montrera également le risque à utiliser des variables globales. Une **variable globale** est une variable qui peut être utilisée (lue ou modifiée) par plusieurs entités (procédure, fonction, programme principal...).

On souhaite écrire un algorithme pour la table traçante qui dessine un carré dont le côté a une longueur de 4 cases. On propose l'algorithme de la figure 1. Exécuter l'algorithme et indiquer quel est le dessin qui apparaît sur la feuille. (On suppose que la table fournit les opérations élémentaires pour écrire une répétition.)

Le problème vient de ce que dans progresser, nous avons oublié de déclarer la variable locale *i*. C'est donc la variable globale (celle du programme principal) de même nom qui est utilisée. Après l'exécution de la procédure progresser, la valeur de la variable *i* du programme principal ne vaut plus 1 mais 5 ! Le **Pour** s'arrête donc (*i* vaut 6 après l'incréméntation) !

**Attention :** En algorithmique, nous déclarons les variables du programme principal juste devant le **Début** du programme principal. Ces variables ne sont donc pas connues des procédures mais seulement du programme principal.

## Listing 1 – Algorithme pour tracer un carré (avec erreur)

---

```

1  Algorithme Tracer_carre
2
3      -- Algorithme pour tracer un carré (avec une erreur !).
4
5  Variables
6      i : Entier  -- pour déplacer le robot sur chaque côté du carré
7
8      -- i est une variable globale car elle est potentiellement
9      -- accessible des deux procédures et du programme principal
10
11 Procédure tourner_gauche Est
12     -- Faire tourner le robot à gauche
13 Variables
14     i : Entier      -- compter le nombre de fois où on tourne
15 Début
16     Pour i <- 1 JusquÀ i = 3 Faire
17         TOURNER
18     FinPour
19 Fin
20
21 Procédure progresser(n : in Entier) Est
22     -- Faire avancer le stylet de n cases
23     --
24     -- Nécessite :
25     --     n > 0 -- n doit être un entier strictement positif
26 Début
27     Pour i <- 1 JusquÀ i = n Faire
28         AVANCER
29     FinPour
30 Fin
31
32 ----- programme principal -----
33 Début
34     BAISSER
35     Pour i <- 1 JusquÀ i = 4 Faire
36         tourner_gauche
37         progresser(4)
38     FinPour
39     LEVER
40 Fin.

```

---



## 5 Structure d'un algorithme

Maintenant que nous avons vu les procédures et les fonctions, nous pouvons donner la structure générale d'un algorithme et l'enchaînement des différentes rubriques qui le composent.

```

1  Algorithme <Nom_algo>
2
3  Constantes      -- Définition des constantes
4      <NOM_CONSTANTE> = <valeur_constante>      -- ce qu'elle représente
5      ...
6
7  Types           -- Définition des types utilisés dans l'algorithme
8      -- explication du type
9      <Type> = <définition du type>
10     ....
11
12 Variable
13     -- Variables globales car potentiellement accessibles par les sous-programmes
14     <nom_variable> : <Type>      -- ce que représente la variable
15     ...
16
17 -- Définition des sous-programmes dans n'importe quel ordre pourvu qu'il n'y
18 -- ait pas de déclarations en avant
19
20 Procédure <verbe> <paramètres_formels> Est
21     -- description de ce que fait la procédure
22     Variable      -- variables locales à la procédure
23     <nom_var_locale> : <Type>      -- ce que représente la variable
24     Début
25     <instruction>      -- élémentaire ou sous-programme
26     ...
27     Fin
28
29 Fonction <nom_fonction> <paramètres_formels> : <Type retour> Est
30     -- description de ce que calcule la fonction
31     Variable      -- variables locales à la fonction
32     ...
33     Début
34     ...
35     Résultat <- ...      -- variable prédéfinie pour toute fonction
36     Fin
37
38 ----- programme principal -----
39
40 Variable      -- variables du programme principal
41     <nom_variable> : <Type>      -- ce qu'elle représente
42 Début
43     <instruction>
44     ...
45 Fin.

```

## 6 Communication entre programmes et sous-programmes

Les sous-programmes communiquent avec le programme principal. La première communication est l'appel de sous-programme lui-même (par exemple `tourner_gauche` ou `tourner_gauche`). Cependant, des informations doivent être échangées entre les sous-programmes et le programme (par pour faire progresser le robot, il faut préciser de combien de cases on peut qu'il avance). Il existe deux façons de transmettre cet information :

- les **paramètres des sous-programmes** : c'est la « bonne » façon de faire. Le programme fournit explicitement les données au sous-programme lors de son appel. Le sous-programme est alors assimilable à une boîte noire dont l'effet ne dépend que de ses paramètres. Ceci favorise la compréhension, l'évolution, la réutilisation, etc.
- les **variables globales** : c'est la mauvaise façon de faire. Les variables globales sont des variables qui sont accessibles à la fois du programme et du sous-programme (variables partagées). Ainsi le programme peut modifier la valeur d'une variable globale, lancer le sous-programme qui utilisera cette variable globale et, enfin, lire le contenu de la variable globale.

Cette façon de faire est très mauvaise car :

- *elle limite la réutilisation* : la fonction est très dépendante de son contexte (les variables globales).
- *la compréhension du programme est difficile* : le passage des paramètres, l'appel à la fonction et la récupération des résultats se fait en plusieurs lignes pas forcément proches (consécutives).
- *le programme est difficile à maintenir* car le résultat d'une fonction ne dépend plus seulement de ses paramètres mais également des valeurs des variables globales qu'elle utilise. Ce sont des entrées déguisées, masquées. Modifier à un endroit la valeur d'une variable globale peut modifier le comportement de plusieurs appels au sous-programme.

### 6.1 Les paramètres

**Paramètres formels** : Lors de la déclaration d'un sous-programme on ne connaît pas encore les données effectives. De plus la généralité d'un sous-programme est de pouvoir travailler sur un ensemble de données qui ont la même forme, la même structure ou les mêmes propriétés. Ces données sont alors référencées sous la forme de paramètres formels.

Une façon de lire l'entête d'un sous-programme est de dire : étant donnés les paramètres...

**Paramètres effectifs** : Les paramètres effectifs sont les données qui sont fournies lors de l'appel d'un sous-programme et qui donc correspondent aux paramètres formels du sous-programme.

Les paramètres (formels, puis effectifs) représentent de l'information qui est échangée entre le programme et le sous-programme. Pour préciser les règles de cette communication, nous définissons des modes de passages de ces paramètres. Un mode de passage définit comment sont transmises les données au sous-programme et les effets des actions que le sous-programmes peut faire dessus.

Jusqu'à présent, nous n'avons vu qu'un seul mode de passage des paramètres : le mode en entrée noté **in**. Dans ce paragraphe, nous commençons par justifier la nécessité d'avoir plusieurs

modes de passage des paramètres et nous présentons ensuite ces différents modes. Nous terminons en décrivant les modes de passage de paramètres en Pascal.

### 6.1.1 Pourquoi des modes de passage ?

Lors de la présentation du langage algorithmique, nous avons vu des opérations élémentaires d'entrée/sortie qui fonctionnent sur les types élémentaires. En fait, ces opérations élémentaires **Lire** et **Écrire** sont tout simplement des procédures prédéfinies par le langage et stockées dans une bibliothèque standard<sup>1</sup>. D'après ce que nous savons des procédures, la définition de **Lire** et **Écrire** pour les entiers pourrait être la suivante :

```
1  Procédure Écrire(n : in Entier) Est
2      -- Afficher à l'écran l'entier n dans sa représentation en base 10
3  Début
4      -- Les instructions nécessaires
5  Fin
6
7  Procédure Lire(n : out Entier) Est
8      -- Lire au clavier les caractères qui doivent correspondre à la
9      -- représentation en base 10 d'un entier, reconstituer l'entier et le
10     -- ranger dans la variable n pour le transmettre au programme appelant.
11  Début
12     -- Les instructions nécessaires
13  Fin
```

**Écrire** reçoit un entier du programme appelant et l'affiche à l'écran. La valeur de l'entier est une information en entrée du sous-programme : elle est fournie par le programme appelant et utilisée par le sous-programme. La paramètre est dit en entrée (pour le sous-programme).

À l'inverse, **Lire** reconstitue un entier à partir des caractères tapés au clavier et transmet l'entier qu'ils représentent au programme appelant. L'information, la valeur de l'entier lu, est donc fournie par le sous-programme au programme appelant.

Si on essaie d'exécuter la procédure **Lire**, on se rend alors compte que la valeur lue ne peut être transmise de la sorte au programme appelant. La valeur de l'entier lu au clavier est en effet rangée dans la zone réservée pour stocker le paramètre formel *n* de la procédure. Or, cette zone n'existe plus après l'exécution de la procédure. Le paramètre est dit en sortie (pour le sous-programme).

### 6.1.2 Quels modes de passage des paramètres ?

Lorsque le programmeur définit une procédure, c'est pour pouvoir ensuite l'utiliser comme une instruction élémentaire.

Nécessairement, il y a un échange de données entre le sous-programme et le programme appelant (programme principal ou sous-programme qui contient un appel au sous-programme considéré). En effet, les calculs effectués par le sous-programme doivent être faits à partir des données du programme appelant. Inversement, des informations calculées dans le sous-programme

---

1. Une bibliothèque regroupe un ensemble de sous-programmes.

pourront ensuite être exploitées par le programme appelant. Pour cette raison, on va caractériser les échanges d'informations en identifiant dans quel sens l'information circule.

Prenons quelques exemples. Lorsque le programmeur utilise la procédure **Écrire** sur les entiers, il lui donne la valeur de l'entier à afficher. On parle alors de paramètre en entrée de la procédure. En revanche, lorsque le programmeur utilise la procédure **Lire** sur les entiers, c'est pour que la procédure lui fournisse une valeur pour la variable qu'il aura passée en paramètre. On parle alors de paramètre en sortie de la procédure. Enfin, une donnée peut être à la fois en entrée et en sortie. On parle alors de paramètre en entrée/sortie.

Pour illustrer l'échange de données entre le sous-programme et le programme appelant, on peut faire une analogie entre un client et un fournisseur (figure 1). Le fournisseur est le sous-programme : il réalise un traitement à la demande du client (le programme appelant). Prenons un exemple concret : vous êtes le client, le fournisseur est un bijoutier. Pour acheter une montre, vous indiquez vos goûts et votre budget (entrées) au bijoutier et il vous vend une montre (sortie). La montre tombe en panne. Vous souhaitez la faire réparer. Vous indiquez au bijoutier les symptômes de la panne (entrée) et vous lui donnez la montre qu'il vous rend ensuite réparée (entrée/sortie).

En algorithmique, nous utiliserons donc ces trois modes : en *entrée*, en *sortie* et en *entrée/sortie*. Nous les détaillons dans les paragraphes suivants.

### 6.1.3 Paramètres en entrée

**Définition :** Un paramètre en entrée correspond à une donnée qui est fournie par le programme appelant au sous-programme. Le sous-programme n'a pas le droit de changer la valeur du paramètre formel, il ne peut qu'accéder à sa valeur.

**Paramètre effectif :** Le paramètre effectif correspondant est n'importe quelle expression dont le type est identique au type du paramètre formel.

**Notation :** Un paramètre formel en entrée est déclaré en utilisant le mot-clé **in** :

<nom> : **in** <Type>      -- paramètre formel correspondant à une donnée en entrée

**Exemple :** La procédure `afficher_mois` affiche le nom correspondant au mois passé en paramètre. L'appelant donne le mois mais n'attend pas de modification en retour. Le mois est donc un paramètre en entrée. (Mois est un type énuméré.)

```

1  Procédure afficher_mois(un_mois : in Mois) Est
2      -- afficher le nom du mois
3      Début
4          Selon un_mois Dans
5              janvier      : Écrire("janvier")
6              février      : Écrire("février")
7              mars         : Écrire("mars")
8              avril        : Écrire("avril")
9              mai          : Écrire("mai")
10             juin         : Écrire("juin")
11             juillet      : Écrire("juillet")

```




FIGURE 1 – Les modes de passage des paramètres

```

12         août         : Écrire("août")
13         septembre   : Écrire("septembre")
14         octobre     : Écrire("octobre")
15         novembre    : Écrire("novembre")
16         décembre    : Écrire("décembre")
17         FinCas
18         Fin

```

#### 6.1.4 Paramètres en sortie

**Définition :** Un paramètre en sortie correspond à une information qui est demandée au sous-programme par le programme appelant.

En conséquence, le sous-programme ne doit pas dépendre (et donc utiliser) la valeur du paramètre formel. De plus, le sous-programme doit impérativement donner une valeur au paramètre effectif.

**Paramètre effectif :** Le paramètre effectif correspondant est nécessairement une variable (ou assimilé : élément d'un tableau ou champ d'un enregistrement) dont le type est identique au type du paramètre formel.

**Notation :** Un paramètre formel en sortie est déclaré en utilisant le mot-clé **out** :

```
<nom> : out <Type>      -- paramètre formel correspondant à une donnée en sortie
```

**Exemple :** La procédure saisir\_mois lit au clavier un mois. Le paramètre mois est donc en sortie. L'utilisateur saisit le mois par son numéro dans l'année. (Mois est un type énuméré.)

```

1  Procédure saisir_mois(un_mois : out Mois) Est
2      -- saisir un mois au clavier. L'utilisateur entre le mois sous la
3      -- forme d'un entier compris entre 1 et 12.
4  Variables
5      numéro : Entier -- numéro du mois saisi au clavier

```

```

6      Début
7          -- Saisir le numéro du mois
8          Lire(numéro);
9
10         -- Traiter les erreurs de saisie
11         TantQue(numéro < 1 Ou numéro > 12) Faire          -- Saisie incorrecte
12             -- Signaler erreur
13             Écrire("Erreur_de_saisie.")
14             -- Saisir un nouveau numéro
15             Écrire("Entrez_un_entier_compris_entre_1_(janvier)_et_12_(décembre)_:_" )
16             Lire(numéro)
17         FinTQ
18
19         -- Convertir le numéro en mois
20         Selon numéro Dans
21             1 : un_mois <- janvier
22             2 : un_mois <- février
23             3 : un_mois <- mars
24             4 : un_mois <- avril
25             5 : un_mois <- mai
26             6 : un_mois <- juin
27             7 : un_mois <- juillet
28             8 : un_mois <- août
29             9 : un_mois <- septembre
30             10 : un_mois <- octobre
31             11 : un_mois <- novembre
32             12 : un_mois <- décembre
33         FinCas
34     Fin

```

**Attention :** Ce qui peut choquer est que, lors d'un appel à la procédure, il faut fournir un paramètre. On pourrait donc se dire que c'est aussi une entrée. Ce n'est pas le cas car, la valeur de cette variable n'est pas utilisée par la procédure. La variable est simplement un contenant qui sert à transmettre une information de la procédure vers l'appelant. C'est comme un client qui va acheter du vin avec un cubitainer. Le cubitainer n'est pas une entrée pour le détaillant, mais seulement le moyen de transmettre le vin au client. Le cubitainer est la variable, le vin (son contenu) est le paramètre en sortie.

### 6.1.5 Paramètres en entrée/sortie

**Définition :** Un paramètre en entrée/sortie correspond à une information à la fois en entrée et en sortie. La procédure doit donc tenir compte de la valeur du paramètre effectif (et d'autres informations éventuellement) pour donner une nouvelle valeur à ce paramètre.

**Paramètre effectif :** Le paramètre effectif correspondant est nécessairement une variable (ou assimilé : élément d'un tableau ou champ d'un enregistrement) dont le type est identique au type du paramètre formel.

**Notation :** Un paramètre formel en entrée/sortie est déclaré en utilisant les mots-clés **in out** :

```
<nom> : in out <Type>  -- paramètre formel correspondant à une donnée
```

-- en entrée-sortie

**Exemple :** La procédure `permuter` est un exemple de procédure qui a des paramètres en entrée-sortie. La valeur de chaque paramètre est utilisée pour modifier la valeur de l'autre paramètre.

```

1  Procédure permuter(a : in out Entier ; b : in out Entier) Est
2      -- échanger les valeurs de deux valeurs entières
3      Variables
4          tmp : Entier      -- utilisé pour faire la permutation
5      Début
6          tmp <- a
7          a <- b
8          b <- tmp
9      Fin

1  Variable
2      i, j : Entier;
3  Début
4      Lire(i)
5      Lire(j)
6      permuter(i, j)      -- OK
7      permuter(i, 5)     -- NON : le deuxième paramètre n'est pas une variable
8  Fin.
```

Sur l'exemple de cette procédure, on constate bien que le sous-programme commence par lire la valeur du paramètre formel (récupère l'information transmise par l'appelant) avant d'en changer la valeur (modification qui sera visible de l'appelant).

**Exemple :** Sous-programme qui incrémente un numéro de mois.

```

1  Procédure incrémenter_mois(num_mois: in out 1..12) Est
2      -- incrémenter le numéro du mois
3      --
4      -- Nécessite
5      --          num_mois >= 1 Et num_mois <= 12
6  Début
7      Si num_mois < 12 Alors
8          num_mois <- num_mois + 1
9      Sinon
10         num_mois <- 1
11     FinSi
12  Fin
```

## 6.2 Les variables globales

**Attention :** L'utilisation des variables globales, même si elle constitue une possibilité pour faire communiquer des sous-programmes, **ne doit pas être utilisée**.

Nous pouvons apporter plusieurs justifications :

- les variables globales rendent le sous-programme dépendant de son contexte. Il ne peut donc plus être réutilisé librement ;

- l'utilisation de variables globales rend la compréhension du sous-programme plus difficile car il ne dépend pas seulement des informations qui lui sont transmises explicitement par les paramètres formels mais également de la valeur (de l'état) des variables globales dont les modifications sont difficiles à prévoir (il faut lire tout le programme) ;
- l'utilisation des variables globales peut conduire à des erreurs très difficiles à trouver (voir l'exemple donné à la section 4).

Il y a cependant des cas où les variables globales sont effectivement la bonne façon de traiter un problème. Des précautions doivent être prises, en particulier de telles variables doivent être isolées du reste de l'application et sont généralement définies à l'intérieur d'un module.

Un exemple où l'utilisation d'une variable globale est nécessaire, c'est lorsqu'une procédure doit conserver une information entre deux appels consécutifs. Par exemple, le nombre de fois où la procédure est appelée. Ceci peut être réalisé par une variable globale qui ne sera utilisée que par cette procédure.

Un autre exemple est lorsqu'on veut représenter une information qui est externe au programme (qui a donc la durée de vie du programme) telle que le nombre de lignes utilisées sur l'écran (pour gérer un affichage page par page). Là encore, il est souhaitable que la variable globale ne soit accessible que par un nombre restreint de sous-programmes.

**Règle :** Même si la plupart des langages de programmation le permettent, une procédure ne doit pas utiliser de variables globales. Ceci est source d'erreurs difficiles à détecter et nuit à la compréhension et donc à la maintenance des programmes. Toutes les informations que manipule un sous-programme doivent donc lui être passées par l'intermédiaire de ses paramètres.

Si on a vraiment besoin de variables globales, alors il faut clairement le préciser dans le commentaire de la variable globale (lister tous les sous-programmes qui l'utilisent et/ou la modifient) et dans le commentaire de tous les sous-programmes qui l'utilisent. La variable globale est en fait un paramètre caché.



## 7 Exécution d'un appel à un sous-programme

### 7.1 Vérification statique

Avant d'être exécuté, un algorithme doit être vérifié. En ce qui concerne l'appel d'un sous-programme, il faut vérifier que :

1. l'appel du sous-programme apparaît à un *endroit autorisé* :
  - dans le cas d'une procédure, l'appel doit correspondre à une instruction ;
  - dans le cas d'une fonction, l'appel doit correspondre à une expression (dont le type est compatible avec le type de retour de la fonction) ;
2. le *nombre de paramètres effectifs* correspond au nombre de paramètres formels ;
3. le *type de chaque paramètre effectif* est identique au type du paramètre formel correspondant.
4. le paramètre effectif est compatible avec le *mode de passage du paramètre formel* :
  - pour un paramètre en entrée : toute expression (du bon type) est autorisée ;
  - pour un paramètre en sortie ou en entrée/sortie : le paramètre effectif doit nécessairement être une variable (ou un élément d'un tableau ou un champs d'un enregistrement) dont le type est identique à celui du paramètre formel.

### 7.2 Exécution d'un appel

Lorsque l'appel d'un sous-programme est statiquement correct, il peut être appelé. Un *bloc d'activation* est créé dans la pile. Il contient les informations suivantes :

- le numéro d'instruction à exécuter après le sous-programme (retour après le déroutement) ;
- les paramètres formels initialisés en fonction des paramètres effectifs ;
- les variables locales du sous-programmes.

Les étapes correspondant à l'exécution de l'appel d'une procédure sont les suivantes :

1. Évaluer les paramètres effectifs qui correspondent à un paramètre formel en entrée.
2. Conserver le numéro de l'instruction qui a provoqué l'appel au sous-programme.
3. Réserver de la place pour les paramètres formels.
4. Initialiser, en fonction de leur mode de passage, les paramètres formels à partir des paramètres effectifs.
  - paramètre en entrée : le paramètre formel est initialisé avec la valeur du paramètre effectif ;
  - paramètre en sortie : le paramètre formel est initialisé avec l'adresse du paramètre effectif. Pour indiquer que la valeur du paramètre effectif n'est pas utilisée, on peut l'initialiser avec un point d'interrogation. Tout accès au paramètre formel est équivalent à un accès au paramètre effectif ;
  - paramètre en entrée/sortie : le paramètre formel est initialisé avec l'adresse du paramètre effectif. Tout accès au paramètre formel est équivalent à un accès au paramètre effectif.

5. Réserver de la place pour les éventuelles variables locales. Elles ne sont pas initialisées. On l'indique en mettant un point d'interrogation.
6. Exécuter les instructions du corps de la procédure comprises entre les mots-clés **Début** et **Fin**.
7. Libérer (effacer) la place occupée par les variables locales et les paramètres formels lorsque l'exécution de la procédure est terminée (le mot-clé **Fin** est atteint).
8. Continuer l'exécution dans le programme appelant (fin du déroutement). L'instruction suivante à exécuter est celle qui suit l'instruction dont le numéro a été conservé lors de l'appel du sous-programme.

**Remarque :** Dans le cas d'une fonction, les paramètres sont forcément en entrée et on ajoute à la liste des variables locales une variable prédéfinie de nom **Résultat**. Sa valeur à la fin de l'exécution du sous-programme (juste avant qu'elle ne soit effacée) correspond à la valeur renvoyée par la fonction.

### 7.3 Illustration

Considérons la procédure suivante :

```

1  Procédure illustrer_exécution(  a : in Entier ;
2                                b : out Entier ;
3                                c : in out Entier) Est
4      Variable
5          i : Entier
6      Début
7          Pour i <- 1 Jusqu'À i = a Faire
8              b <- b + a
9          FinPour
10         c <- b
11      Fin

```

et l'appel :

```

12  Variable
13      n1, n2: Entier
14  Début
15      n1 <- 1
16      n2 <- 2
17      illustrer_exécution(3, n1, n2)
18  Fin.

```

La figure 2 indique l'état de la pile correspondant à l'exécution de ce programme jusqu'à l'appel de la procédure (les étapes 1 à 5 ont été réalisées, l'instruction suivante à exécuter est donc celle qui porte le numéro 7, première instruction de la procédure). L'étape suivante est donc la 6 : exécuter les instructions du corps de la procédure.




FIGURE 2 – État de la mémoire avant l'exécution du corps de la procédure

## 8 Procédures

Dans les paragraphes précédents, nous avons introduit la notion de procédure en partant d'exemples. Dans ce paragraphe, nous donnons une définition plus précise d'une procédure. En particulier, nous insistons sur les notions d'interface et d'implantation d'une procédure. L'interface s'adresse à l'utilisateur de la procédure : elle décrit ce que fait la procédure et la manière de l'utiliser. L'implantation s'adresse au processeur : elle décrit l'enchaînement d'instructions correspondant au traitement réalisé par la procédure. Nous terminons ce paragraphe en décrivant comment définir une procédure.

### 8.1 Interface d'une procédure

L'interface s'adresse à l'utilisateur de la procédure (un programmeur). Elle doit donc préciser la syntaxe d'appel de la procédure : c'est la signature ou le prototype de la procédure. Cependant, la connaissance de la syntaxe d'appel n'est pas suffisante. Il est également nécessaire de savoir ce que fait la procédure : c'est la sémantique de la procédure qui, malheureusement, ne peut souvent qu'être décrite de manière informelle dans un commentaire<sup>2</sup>.

#### 8.1.1 Signature

Pour définir la syntaxe d'appel, il est nécessaire de définir :

- le **nom** de la procédure. Ce nom doit être significatif, c'est-à-dire qu'il doit suggérer le traitement que réalise la procédure. Puisqu'une procédure correspond à un traitement, une instruction de haut niveau, on lui donnera<sup>3</sup> pour nom un verbe à l'infinitif ;
- les **paramètres formels** de la procédure. Un paramètre formel décrit une information échangée entre le programme appelant (à l'origine de l'appel) et la procédure. Il est com-

2. La programmation contractuelle telle qu'elle est proposée par B. Meyer est une solution partielle à ce problème qui n'est malheureusement pas supportée par les langages de programmation traditionnels.

3. pourra lui donner. Ceci est conseillé mais, bien entendu, pas obligatoire !

posé d'un nom (significatif bien sûr), d'un type (tout type est possible) et d'un mode de passage (en entrée, en sortie ou en entrée/sortie, voir paragraphe 6.1).

Une procédure peut avoir zéro, un ou plusieurs paramètres formels. Les différents paramètres formels sont séparés par des points virgules.

**Conseil :** Une procédure ne doit pas avoir trop de paramètres. Si le nombre de paramètres est trop important, c'est peut-être parce que la procédure fait trop de choses, il faut alors la décomposer, peut-être parce qu'elle est trop élémentaire, peut-être parce que les données manquent de structure, etc.

### 8.1.2 Sémantique

La sémantique d'une procédure est décrite dans un commentaire. Ce n'est pas parce que le compilateur ne peut pas exploiter les commentaires qu'il faut les négliger. En effet, les programmes sont avant tout lus lors du développement initial mais surtout pour les phases de maintenance (correction et évolution) par des êtres humains (donc des êtres capables de lire et comprendre les commentaires).

Le commentaire doit faire apparaître les informations suivantes :

- l'**objectif** de la procédure. Il doit décrire ce que fait la procédure et non comment elle le fait. Cette description doit être précise et succincte ;
- les **préconditions** sur les paramètres formels en entrée. Elles décrivent les propriétés que doivent respecter les paramètres en entrée pour que la procédure puisse être exécutée ;
- les **postconditions** sur les paramètres en sortie : elles décrivent les propriétés qui caractérisent les paramètres en sortie après l'exécution de la procédure.

Les rubriques ci-dessus sont essentielles. Les renseigner systématiquement aide à bien définir ce que fait la procédure.

Il est possible de rajouter d'autres rubriques comme par exemple la complexité de la procédure (un ordre de grandeur du temps d'exécution de la procédure en fonction de la valeur des paramètres en entrée). On peut également indiquer la solution retenue pour implanter la procédure, mais sa description doit être très succincte.

## 8.2 Implantation d'une procédure

L'interface décrit ce que fait la procédure. L'implantation décrit comment elle le fait. Elle s'adresse donc au processeur. En d'autres termes, l'interface décrit le problème à résoudre, l'implantation décrit une solution (un algorithme) pour le résoudre. Comme tout algorithme, il est donc composé de **variables locales** et d'**instructions**.

## 8.3 Comment définir une procédure ?

Ici, nous proposons une manière de définir une procédure. Ce n'est bien entendu pas la seule manière de faire. Il est important de commencer par définir l'interface (savoir ce que fait la procédure) avant l'implantation (savoir comment elle le fait). Nous conseillons donc de suivre les étapes suivantes :

**1. Définir l'interface de la procédure :**

- (a) Définir l'*objectif* de la procédure. Ceci a pour objectif de bien comprendre ce que doit faire la procédure (cf R0 des raffinages).
- (b) Identifier les *paramètres formels* et leur mode de passage (entrée et/ou sortie).
- (c) En déduire un *nom significatif* pour la procédure.
- (d) Identifier les *préconditions* et les *postconditions* sur ces paramètres (et l'état du système).
- (e) Rédiger l'interface de la procédure à partir des informations ci-dessus.

2. **Définir l'implantation de la procédure :** Une fois l'interface définie, on peut alors s'intéresser à son **implantation**. Une méthode possible est celle par raffinages successifs, le R0 étant défini par l'interface de la procédure. On a en effet ce que doit faire la procédure (objectifs et postconditions), les données manipulées (paramètres formels) et les hypothèses sur ces paramètres (préconditions).

**Remarque :** Une fois l'implantation terminée, on peut compléter l'interface en indiquant la complexité de la solution retenue, le principe de la solution, etc.

## 9 Fonctions

Tout comme une procédure, une fonction est un sous-programme. La différence est qu'une fonction renvoie une information et donc peut être assimilée à une expression, alors qu'une procédure effectue un traitement et correspond donc à une instruction. Comme une procédure, une fonction a une interface et une implantation que nous décrivons ci-dessous.

### 9.1 Interface d'une fonction

L'interface d'une fonction est composée d'une partie syntaxique correspondant à sa signature et d'une partie sémantique.

#### 9.1.1 Signature

La signature d'une procédure est composée :

- d'un **nom** (significatif) qui caractérise l'information renvoyée par la fonction et est donc soit un nom commun, soit un adjectif<sup>4</sup> ;
- des **paramètres formels**. Ils sont tous en entrée. Une fonction ne peut pas avoir de paramètres en sortie ou en entrée/sortie. Une fonction doit toujours avoir au moins un paramètre formel.

**Justification** : Si une fonction n'a pas de paramètres, c'est une constante ! Sinon c'est qu'elle utilise une variable globale, ce qui est à éviter, ou qu'elle réalise des entrées/sorties. Dans ce dernier cas, il faudrait en faire une procédure, comme **Lire** ;

- un **type de retour** qui est le type de l'information renvoyée.

#### 9.1.2 Sémantique

La sémantique d'une fonction est précisée dans un commentaire qui contient les rubriques suivantes :

- l'**objectif** de la fonction qui décrit ce que calcule la fonction ;
- les **préconditions** sur les paramètres formels ;
- les **(post)conditions** sur le résultat de la procédure.

Comme dans le cas d'une procédure, d'autres rubriques peuvent être ajoutées.

### 9.2 Implantation d'une fonction

L'implantation d'une fonction décrit comment le calcul est réalisé. Elle est similaire à celle d'une procédure à la différence qu'une variable locale à la fonction est automatiquement déclarée. Son nom est **Résultat** et son type est le type de retour de la fonction. Elle peut (doit) être utilisée pour réaliser les calculs. La valeur renvoyée par la fonction est la valeur de la variable **Résultat** à la fin de l'exécution des instructions de la fonction.

---

4. Ceci est conseillé mais, bien entendu, pas obligatoire !

## 10 Procédure ou fonction ?

Une fonction peut être vue comme un cas particulier de procédure. Dans les paragraphes suivants, nous expliquons comment réécrire une fonction sous la forme d'une procédure. Nous en déduisons l'intérêt de définir des fonctions plutôt que des procédures. Enfin, nous définissons des critères permettant de décider si un sous-programme doit être décrit comme une fonction ou une procédure.

### 10.1 La fonction comme procédure

Toute fonction peut s'écrire sous la forme d'une procédure. La procédure correspondante prend un paramètre formel de plus que la fonction. Donnons un nom à ce paramètre supplémentaire... au hasard... résultat. Son type est le type de retour de la fonction et son mode de passage est en sortie. Le nom de la procédure est le nom de la fonction précédée de « calculer\_ ».

Prenons un exemple concret : la fonction max qui renvoie le plus grand de deux entiers.

```

1  Fonction max(a, b : in Entier) : Entier Est
2      -- le plus grand de deux nombres entiers
3      --
4      -- Assure
5      --      (Résultat >= a) Et (Résultat >= b) -- plus grand que a et b
6      --      (Résultat = a) Ou (Résultat = b)   -- l'un des deux
7  Début
8      Si a > b Alors
9          Résultat <- a
10     Sinon
11         Résultat <- b
12     FinSi
13 Fin

```

La procédure correspondante s'écrit alors :

```

1  Procédure calculer_max(résultat: out Entier ; a : in Entier ; b : in Entier) Est
2      -- calculer dans résultat le plus grand des deux nombres entiers a et b
3  Début
4      Si a > b Alors
5          résultat <- a
6      Sinon
7          résultat <- b
8      FinSi
9  Fin

```

**Remarque :** Dans l'optique d'un sous-programme le paramètre en sortie que nous avons appelé résultat car il correspond au **Résultat** de la fonction devrait s'appeler max, comme la fonction. Ce n'est pas un hasard !

### 10.2 Intérêt des fonctions par rapport aux procédures

Si la fonction max et la procédure calculer\_max sont strictement équivalentes par rapport à ce qu'elles calculent, la fonction est d'un emploi beaucoup plus souple. Prenons un exemple. On

veut calculer la plus grande des valeurs de 3 variables entières (n1, n2 et n3) et de la constante entière 15 et mettre le résultat dans la variable `plus_grand`. On suppose les différentes variables déjà déclarées. En utilisant la fonction `max`, on peut écrire :

```
plus_grand <- max(max(n1, n2), max(n3, 15))
```

alors qu'avec la procédure on est obligé de décomposer les calculs en écrivant (on suppose `pg1` et `pg2` deux variables entières déjà déclarées) :

```
calculer_max(pg1, n1, n2)
calculer_max(pg2, n3, 15)
calculer_max(plus_grand, pg1, pg2)
```

Ce qui peut également s'écrire :

```
calculer_max(plus_grand, n1, n2)
calculer_max(plus_grand, plus_grand, n3)
calculer_max(plus_grand, plus_grand, 15)
```

On constate que l'écriture avec la fonction est plus naturelle et que le calcul peut s'effectuer en une ligne grâce à une expression composée. Bien entendu, il faut éviter d'écrire des expressions trop compliquées qui deviendraient difficiles à lire et à comprendre.

L'intérêt des fonctions par rapport aux procédures est donc d'ordre syntaxique : elles sont plus facile à utiliser car elles peuvent apparaître dans une expression (et si on veut les utiliser pour initialiser une variable, il suffit de les utiliser avec l'affectation).

### 10.3 Choisir entre fonction et procédure

Puisqu'une fonction est plus souple d'utilisation, il est préférable d'écrire une fonction plutôt qu'une procédure. Cependant, ceci n'est pas toujours possible. Des contraintes présentées sur le type de retour et les paramètres d'une fonction (section 9), on peut en déduire la règle suivante.

**Règle pour pouvoir écrire une fonction :** Si on a identifié un sous-programme qui a les propriétés suivantes :

- elle a un seul paramètre formel en sortie,
- tous ses autres paramètres formels (il y en a au moins un) sont en entrée,
- elle ne réalise ni saisie, ni sortie,

alors il est préférable de l'écrire sous la forme d'une fonction.

**Remarque :** Un sous-programme d'initialisation est souvent traduit en procédure même si une fonction serait également possible.

**Conseil :** Il faut donc, dans un premier temps, raisonner en terme de sous-programme. C'est lorsque vous aurez identifié les paramètres formels (et leur mode de passage) que vous pourrez choisir de décrire le sous-programme comme une fonction ou une procédure (suivant que la règle ci-dessus s'applique ou non).



## 11 Liens entre raffinages et sous-programmes

Chaque niveau intermédiaire d'un raffinement est un candidat à être une procédure (ou une fonction pour les expressions). Ce n'est cependant pas une condition nécessaire. Tout niveau intermédiaire ne doit pas être transformé en sous-programme.

Les critères qui font qu'un niveau de raffinement est transformé en sous-programme sont :

- le *sous-programme est réutilisable* : est-ce que le traitement est général et pourra être utilisé à plusieurs endroits dans l'algorithme ou éventuellement dans d'autres algorithmes ?
- l'*amélioration de la lisibilité* : le sous-programme représente quelques lignes de code qui, si elle étaient laissées dans le programme appelant, le rendrait difficile à lire (trop long, trop détaillé, etc.) ;
- le *sous-programme correspond à un traitement bien défini*, relativement indépendant (pas trop spécifique) des autres étapes de l'algorithme.

En particulier, le nombre de paramètres de la procédure doit être limité. Si une procédure a un grand nombre de paramètres, c'est qu'elle est trop liée à son contexte d'appel. En faire une procédure ne permettra certainement pas de gagner en clarté.

## 12 Imbrication de sous-programmes

Chaque niveau de raffinement intermédiaire peut correspondre à un sous-programme (paragraphe 11). Par ailleurs, chaque sous-programme constitue un sous-problème qui pour être résolu doit être raffiné (paragraphe 8). Ainsi, le raffinement d'une procédure peut faire intervenir des étapes de haut niveaux qui, à leur tour, peuvent correspondre à des sous-programmes. Puisque ces sous-programmes ont été identifiés pendant le raffinement d'un sous-programme, ils peuvent être considérés comme spécifiques, locaux à ce sous-programme. On parle alors de **sous-programmes imbriqués**.

Tout comme une variable locale d'un sous-programme, un sous-programme local à un sous-programme ne sera visible (et donc utilisable) que depuis ce sous-programme. Si son nom correspond à celui d'une entité définie à un niveau supérieur, il masque cette entité.

En algorithmique, nous déclarerons les sous-programmes locaux à un sous-programme avant la déclaration des variables locales du sous-programme englobant. Tout comme dans le cas de l'algorithme principal, ceci évite que les sous-programmes utilisent par inadvertance ces variables locales. Attention cependant, les paramètres formels du sous-programme englobant sont des **données globales** pour les sous-programmes locaux. Il faut donc bien faire attention à ce qu'on fait.

### 12.1 Tolérance sur les sous-programmes

Nous avons dit dans le paragraphe 8, qu'un sous-programme ne devait manipuler que les données qui lui sont passées en paramètres. En particulier, il ne doit pas utiliser de variables globales. Cependant, dans le cas de sous-programmes imbriqués, il est toléré qu'ils puissent utiliser les paramètres formels (et même des variables !) du sous-programme englobant. Ceci est justifié par le caractère local et donc spécifique du sous-programme.

**Conseil :** Je pense toutefois que ce genre de pratique est à éviter car il conduit à des erreurs difficiles à détecter (voir l'exemple de la section 4) et que d'autre part les sous-programmes sont plus difficiles à comprendre et à maintenir car ils dépendent de leur contexte (du sous-programme englobant et de ses données).

### 12.2 Sous-programmes d'intérêt général

Si, lors du raffinement d'un sous-programme, vous identifiez un sous-programme qui est suffisamment général pour pouvoir être utilisé à d'autres endroits de l'algorithme ou dans de futurs algorithmes, il est souhaitable de ne pas en faire un sous-programme local, mais de le faire apparaître au premier niveau.

Ainsi, s'il y a effectivement un lien entre raffinement et sous-programme, on constate toutefois que la structure hiérarchique du raffinement (dédite de l'analyse et de la résolution du problème posé) peut ne pas correspondre à la structure hiérarchique d'imbrication des sous-programmes (des sous-programmes apparus relativement tard dans le raffinement peuvent apparaître au premier niveau dans l'algorithme final).

Reprenons l'exemple du paragraphe 2.5. Pour implanter la procédure `reculer` qui fait reculer le robot de  $n$  cases, nous identifions le raffinement composé des étapes suivantes :

- 1 faire demi-tour
- 2 progresser de  $n$  cases
- 3 faire demi-tour

Chacune de ces étapes est elle-même complexe et peut donc donner lieu à une procédure. Comme elles sont identifiées au cours du raffinement de la procédure `reculer`, elles peuvent être définies comme procédures locales. Voici l'algorithme correspondant :

```

1  Procédure reculer(n : in Entier) Est
2      -- faire avancer le robot de n cases vers l'arrière, n étant
3      -- strictement positif. La direction du robot n'est pas modifiée.
4      --
5      -- Algorithme : avancer dans la direction opposée.
6
7      Procédure faire_demi_tour Est
8          -- faire faire demi tour au robot
9          Début
10             PIVOTER
11             PIVOTER
12          Fin
13
14      Procédure progresser(n : in Entier) Est
15          -- Faire avancer le robot de n cases dans la direction courante.
16          -- N doit être un entier strictement positif
17          Variable
18             i : Entier
19          Début
20             Pour i <- 1 Jusqu'À i = N Faire
21                 AVANCER
22             FinPour
23          Fin
24
25      Début          -- procédure reculer
26          faire_demi_tour
27          progresser(n)          -- n est le paramètre effectif de progresser
28          faire_demi_tour
29      Fin

```

Cependant les procédures `faire_demi_tour` et `progresser` sont suffisamment générales pour être utilisées dans d'autres algorithmes. Aussi, au lieu de les définir localement, il est préférable de les définir au premier niveau. Elles pourront ainsi être réutilisées.

Noter que d'imbriquer les sous-programmes n'améliore pas la lecture du code. En conséquence, il faut éviter d'avoir plusieurs niveaux d'imbrication et/ou trop de sous-programmes locaux.

### 12.2.1 Sous-programmes et langages de programmation

Si l'imbrication est possible dans certains langages (Pascal, Modula2, Ada, Lisp, Caml...), elle est interdite dans d'autres (C, C++, Eiffel, F...).